
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.1.4.8, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_1_4_8
Test Group:
MAC

ATS Version:
iwd-EUTRA-TVB2009-03_D10wk11
System Simulator used:
Anite Conformance Toolset
UE used:
LG Electronics LD100U UE, Samsung Tanum UE.

Verification Status:
PASS
4 Corrections required for test case 7.1.4.8

4.1 Introduction

This section describes the changes required to make test case 7.1.4.8 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D10wk11 release.

4.2 Change 1

Testcase name
f_TC_7_1_4_8_EUTRA

Reason for change
In the expected sequence SS should send multiple PDU’s on each DRB and then expect UE to send BSR based on periodic BSR timer expiry and loopback data simultaneously. This makes implementation difficult to differentiate between BSR coming as result of padding BSR and periodic BSR timer expiry. Moreover from the expected sequence it not clear how many grants should be sent to UE at various steps.

Summary of change
Expected sequence is now modified to send 1 RLC SDU at step 2 and 8. Also it is now specified that grants are transmitted each time to the UE based on the UE periodic BSR timer expiry.

TTCN is modified to implement as per the prose CR.

Source of change


Before:

  function f_TC_7_1_4_8_EUTRA ( ) runs on EUTRA_PTC

  { /* L2/MAC

     * 7.1.4.8 : Correct Handling of MAC control information [Buffer Status/ Periodic BSR Timer expires] */

    var integer j, k, l, m;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var boolean v_Flag := true;

    timer t_Watchdog := 5.0;

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_Init (c1);

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    f_EUTRA_CellConfig_DRB_Common (eutra_Cell1, 1, 0);

    //Configure cell with 2 AM DRBs

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm

    // creates cell and performs registration

    // step to configure SS RLC/PDCP and MAC in No header manipulation mode on SS Side for DRB

    f_EUTRA_LoopBackActivation_State4_7148();

    f_SS_ConfigBufferStatusReportIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

    // Configure SS to report Reception of MAC Control elements BSR

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true,layer2); // @sic R5-101050.

    //Step 3: UL Grant of 32 bits

    f_EUTRA_StartExplicitULGrantTransmissionOnSR_Reception (eutra_Cell1,

                                                            cs_TimingInfo_Now,

                                                            0, 2); /* Nprb = 2, Itbs = 0     Table 7.1.7.2.1-1 of 36.213

                                                                    * Imcs = 0 -> Itbs = 0   MCS index of table 8.6.1-1 of 36.213 */

    //Step 2

    //@sic R5s100066

    DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU  (eutra_Cell1,

                                                   tsc_RbId_DRB2,

                                                   cs_TimingInfo_SubFrame(0),

                                                   f_BuildPDCPSDUsToBeSentInSingleTTI( crs_PDCP_SDU_12B, 50 )));

    //Step 4 //R5-095076

    SYSIND.receive(car_BSR_IND(eutra_Cell1, cr_TimingInfo_Any, cr_ShortBSR_Ind(1, (1..63))));  // Any signalling data control PDU's might be possible

    //Step 5: UL Grant of 136 bits

    f_EUTRA_PeriodicCyclicULGrantTransmission (eutra_Cell1,

                                               cs_TimingInfo_Now,

                                               9, 1, 1, 20); /* Nprb = 1, Itbs = 9          Table 7.1.7.2.1-1 of 36.213

                                                              * Imcs = 9 -> Itbs = 9   MCS index of table 8.6.1-1 of 36.213

                                                              * period of 20 is selected to allow 2 BSR periodic  one in first few sub frames

                                                              as periodic BSR timer is started after step 4*/

    j:=0;

    k:=0;

    t_Watchdog.start;

    alt {

      [] SYSIND.receive(car_BSR_IND(eutra_Cell1, cr_TimingInfo_Any, cr_ShortBSR_Ind(1, (1..63))))

        { // Check BSR value > 0

          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "");

          j := j+1;

          if (j<3) //R5-101067

            {

              repeat;

            }

          else

            {

              t_Watchdog.stop;

            }

        }

      [] DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList (eutra_Cell1, tsc_RbId_DRB2, cr_TimingInfo_Any, cr_PDCP_SDUList_1(crs_PDCP_SDU_12B)))

        {

          k:=k+1;

          repeat;

        }

    };

    //Step 7: SS ignores scheduling requests and does not allocate any uplink grant

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync( eutra_Cell1, cs_TimingInfo_Now );

    //Step 8a: UL Grant of 32 bits

    //@sic R5-101067

    f_EUTRA_StartExplicitULGrantTransmissionOnSR_Reception (eutra_Cell1,

                                                            cs_TimingInfo_Now,

                                                            0, 2); /* Nprb = 2, Itbs = 0     Table 7.1.7.2.1-1 of 36.213

                                                                    * Imcs = 0 -> Itbs = 0   MCS index of table 8.6.1-1 of 36.213 */

    //Step 8

    //@sic R5s100066

    DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU  (eutra_Cell1,

                                                   tsc_RbId_DRB1,

                                                   cs_TimingInfo_SubFrame(0),

                                                   f_BuildPDCPSDUsToBeSentInSingleTTI( crs_PDCP_SDU_12B, 30 )));

    // step 8b

    //@sic R5-101067

    t_Watchdog.start;

    SYSIND.receive(car_BSR_IND(eutra_Cell1,cr_TimingInfo_Any, cr_LongBSR_Ind(0, (1..63), (1..63), 0))) ;

    t_Watchdog.stop;

    //Step 9: UL Grant of 136 bits

    f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1,

                                               cs_TimingInfo_Now,

                                               9, 1, 1, (30+2+30+7)); /* Nprb = 1, Itbs = 9          Table 7.1.7.2.1-1 of 36.213

                                                                       * Imcs = 9 -> Itbs = 9   MCS index of table 8.6.1-1 of 36.213

                                                                       * 30 PDU's on LCH 2, 30 PDU's on LCH 1 and two Long BSR.

                                                                       * 7 Additional grants added for any residue left */

    m:=0;

    l:=0;

    // step 10 and 11

    t_Watchdog.start;

    alt {

      [v_Flag] SYSIND.receive(car_BSR_IND(eutra_Cell1,cr_TimingInfo_Any, cr_LongBSR_Ind(0, (1..63), (1..63), 0))) //R5-095076

        { // received UL MAC PDU(Long BSR (BS#2>0;BS#3>0)

          // Check BSR value >0

          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "");

          m := m+1;

          repeat;

        }

      [v_Flag] SYSIND.receive(car_BSR_IND(eutra_Cell1,cr_TimingInfo_Any, cr_LongBSR_Ind(0, 0, (1..63), 0))) //R5-095076

        { // Eat any Long BSR's reported with data on low priority channel

          repeat;

        }

      [v_Flag] SYSIND.receive(car_BSR_IND(eutra_Cell1, cr_TimingInfo_Any, cr_ShortBSR_Ind(2, ?)))

        { // Eat any short BSR's reported

          repeat;

        }

      [v_Flag] SYSIND.receive(car_BSR_IND(eutra_Cell1, cr_TimingInfo_Any, cr_TruncatedBSR_Ind(1, ?)))

        { // Eat any short BSR's reported

          repeat;

        }

      [v_Flag] DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList (eutra_Cell1, tsc_RbId_DRB2, cr_TimingInfo_Any, cr_PDCP_SDUList_1(crs_PDCP_SDU_12B)))

        {

          k:=k+1;

          repeat;

        }

      [v_Flag] DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList (eutra_Cell1, tsc_RbId_DRB1, cr_TimingInfo_Any, cr_PDCP_SDUList_1(crs_PDCP_SDU_12B)))

        {

          l:=l+1;

          repeat;

        }

      [] SYSIND.receive(car_BSR_IND(eutra_Cell1, cr_TimingInfo_Any, cr_LongBSR_Ind(0,0,0,0)))

        {

          v_Flag := false;

          repeat;

        }

      [] t_Watchdog.timeout

        {

          if ((j<3) or (k<50) or (l<30) or (m<2))

            {

              f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE is not sending atleast 2 MAC Long BSR or loop backed PDU's");

            }

          else

            {

              f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "all Loop back PDU's  and Buffer Status reports received");

            }

        }

    };

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    f_SS_ConfigBufferStatusReportIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);

    // Configure SS to report Reception of MAC Control elements BSR

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);

    // To deactivate test loop mode before RRC release

    f_OpenUE_TestLoopMode_Deactivate_TestMode(eutra_Cell1);

    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, RRC_CONNECTED, ATTACHED);

  }; // end of 7.1.4.8

After:

  function f_TC_7_1_4_8_EUTRA ( ) runs on EUTRA_PTC

  { /* L2/MAC

     * 7.1.4.8 : Correct Handling of MAC control information [Buffer Status/ Periodic BSR Timer expires] */

    var integer j, k, l, m;

    var octetstring v_EncodedRlcPdu;

    var octetstring v_EncodedRlcPdu1;

    var octetstring v_EncodedRlcPdu2; 

    var octetstring v_EncodedRlcPdu3;

    var octetstring v_EncodedRlcPdu4;           

    var octetstring v_EncodedPdcpPdu;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var boolean v_Flag := true;

    var SubFrameTiming_Type v_Timing;

    timer t_Watchdog := 5.0;

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_Init (c1);

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    //f_EUTRA_CellConfig_DRB_Common (eutra_Cell1, 1, 0);

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    //Configure cell with 2 AM DRBs

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm

    // creates cell and performs registration

    // step to configure SS RLC/PDCP and MAC in No header manipulation mode on SS Side for DRB

    f_EUTRA_LoopBackActivation_State4_7148();

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1,cs_MAC_TestMode_NoHeaderManipulationDL_UL); 

    // Configure SS to report Reception of MAC Control elements BSR

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    //Step 3: UL Grant of 32 bits

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync (eutra_Cell1, cs_TimingInfo_Now);

    v_EncodedRlcPdu1 := f_RLC_PDU_Encvalue (cs_RLC_AMD_1PDU(tsc_NoPoll,

                                                            int2bit (0,10),

                                                            f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_12B)))));

    v_EncodedRlcPdu2 := f_RLC_PDU_Encvalue (cs_RLC_AMD_1PDU(tsc_NoPoll,

                                                            int2bit (0,10),

                                                            f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_12B1)))));

    v_EncodedRlcPdu3 := f_RLC_PDU_Encvalue (cs_RLC_AMD_1PDU(tsc_Poll,

                                                            int2bit (0,10),

                                                            f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_12B)))));

    v_EncodedRlcPdu4 := f_RLC_PDU_Encvalue (cs_RLC_AMD_1PDU(tsc_Poll,

                                                            int2bit (0,10),

                                                            f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_12B1)))));

    v_Timing := f_EUTRA_GetNextSendOccasion (eutra_Cell1);                                                                  

    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell1,

                                          tsc_RbId_DRB1,

                                          cs_TimingInfo(v_Timing.SFN.Number, 0),

                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB2, v_EncodedRlcPdu1))));

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo((v_Timing.SFN.Number+6)mod 1024, 0),

                                    0, 2); // MCS = 2 NPRB = 1 => 32 bits grant

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo((v_Timing.SFN.Number+10)mod 1024, 0),

                                    0, 2); // MCS = 2 NPRB = 1 => 32 bits grant

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo((v_Timing.SFN.Number+14)mod 1024, 0),

                                    0, 2); // MCS = 2 NPRB = 1 => 32 bits grant                                    

    DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo_Any,

                                        {cr_MAC_PDU_2Padding_ShortBSR('01'B,'000011'B)}));

    DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo_Any,

                                        {cr_MAC_PDU_2Padding_ShortBSR('01'B,'000011'B)}));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "UE is sending short BSR");

    DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo_Any,

                                        {cr_MAC_PDU_2Padding_ShortBSR('01'B,'000011'B)}))

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "UE is sending Short BSR");

    v_Timing := f_EUTRA_GetNextSendOccasion (eutra_Cell1);

    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell1,

                                          tsc_RbId_DRB1,

                                          cs_TimingInfo(v_Timing.SFN.Number, 0),

                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRlcPdu2))));

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo((v_Timing.SFN.Number+6)mod 1024, 0),

                                    0, 2); // MCS = 2 NPRB = 1 => 32 bits grant

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo((v_Timing.SFN.Number+10)mod 1024, 0),

                                    0, 2); // MCS = 2 NPRB = 1 => 32 bits grant

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo((v_Timing.SFN.Number+14)mod 1024, 0),

                                    0, 2); // MCS = 2 NPRB = 1 => 32 bits grant

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo((v_Timing.SFN.Number+16)mod 1024 , 0),

                                    7, 3); 

    DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo_Any,

                                        {cr_MAC_PDU_LongBSR(?,'000011'B,'000011'B,?)}));

    DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                       tsc_RbId_DRB1,

                                       cr_TimingInfo_Any,

                                       {cr_MAC_PDU_LongBSR(?,'000011'B,'000011'B,?)}))

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "UE is sending Long BSR");

    DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                       tsc_RbId_DRB1,

                                       cr_TimingInfo_Any,

                                       {cr_MAC_PDU_LongBSR(?,'000011'B,'000011'B,?)}))

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "UE is sending Long BSR");

    alt

    {

          [] DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                           tsc_RbId_DRB1,

                                           cr_TimingInfo_Any,   

                                           {cr_MAC_PDU_ShortBSR_2Subheader_Short_Padding (?,

                                                                      ?,

                                                                     tsc_LchId_DRB2,int2bit(16, 7), v_EncodedRlcPdu3,

                                                                     tsc_LchId_DRB1,int2bit(16, 7), v_EncodedRlcPdu4,

                                                                     cr_Octet6_Any)}));

           [] DRB.receive( car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                           tsc_RbId_DRB1,

                                           cr_TimingInfo_Any,   

                                           {cr_MAC_PDU_LongBSR_2PaddingSubheader_RLCPDU_7148 (?,

                                                                      ?, ?,?, 

                                                                     tsc_LchId_DRB2,int2bit(16, 7), v_EncodedRlcPdu3,

                                                                     tsc_LchId_DRB1,v_EncodedRlcPdu4)}));                                                           

    }       

    f_RLC_Status_PDU_Tx(tsc_LchId_DRB1,1);

    f_RLC_Status_PDU_Tx(tsc_LchId_DRB2,1);

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    //f_SS_ConfigBufferStatusReportIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);

    // Configure SS to report Reception of MAC Control elements BSR

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1,cs_MAC_TestMode_NoHeaderManipulationDLOnly);

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);

    // To deactivate test loop mode before RRC release

    f_OpenUE_TestLoopMode_Deactivate_TestMode(eutra_Cell1);

    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, RRC_CONNECTED, ATTACHED);

  }; // end of 7.1.4.8

5 Execution Log Files

5.1 LG Electronics LD100U UE

The LG UE passed this test case on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file TC_7_1_4_8_LG-UE.htm:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.2 Samsung Tanum UE

The Samsung Tanum UE passed this test case on the Anite Conformance Toolset solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file TC_7_1_4_8_Samsung-UE.htm:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

[1]
R5s100131: This archive comprises text format execution log file.

